Physics 651: Assignment 3

(to be submitted by Tuesday, October 1, 2024)

I invite you to attempt Assignment 3 and to turn in your work for Questions 1–3. Any hand-written derivations should be submitted to me in hard copy. Any computational results should be collected in a single Wolfram Notebook and sent as an attachment to `kbeach@olemiss.edu`. Please follow the naming convention `Phys651-A3-webid.nb`, and be sure to include the subject line `Phys651-Fall2024-webid Assignment 3 Submission`.

1. The Cauchy-Schwarz inequality states that $(u,v)^2 \le (u,u)(v,v)$ for any compatible vectors u and v in a vector space endowed with an inner product $(\cdot,\cdot)$. Solve these first two problems by hand by applying Cauchy-Schwarz with an appropriate choice of u and v.

   (a) Prove that for any $a, b, c > 0$,

   $$ab + bc + ca \le a^2 + b^2 + c^2.$$

   You may find that this *Mathematica* code helps to clarify your thinking:

```
u = {a, b, c}
v = RotateLeft[u]
(u.v)^2
(u.u) (v.v)
Assuming[a > 0 && b > 0 && c > 0, (u.v)]
Assuming[a > 0 && b > 0 && c > 0, Refine[Sqrt[(u.u) (v.v)]]]
```

   (b) Prove (Nesbitt's Inequality) that for any $a, b, c > 0$,

   $$\frac{a}{b+c} + \frac{b}{a+c} + \frac{c}{a+b} \ge \frac{3}{2}.$$

   The following code may help. If you're really stuck, have a look at this video.

```
u = {Sqrt[b + c], Sqrt[a + c], Sqrt[a + b]}
v = 1/u
Total[Array[#1/(#2 + #3) & @@ RotateRight[{a, b, c}, #] &, 3]] >= 3/2
(u.v)^2
Expand[(u.u)(v.v)]
```

   The following code is *not* a proof, but it will let you verify that the inequality is satisfied for 5 000 random instantiations of triples a, b, c drawn uniformly from the interval $[0, 10\,000)$. Give it a try.

```
vals = Table[{RandomReal[10000], RandomReal[10000], RandomReal[10000]}, {i, 1, 5000}];
f[a_, b_, c_] := a/(b + c) + b/(a + c) + c/(a + b)
And @@ (# >= 3/2 & /@ f @@@ vals)
```

2. A tiny ball of mass m rolls on a curved two-dimensional surface, parameterized by $z = x^2 - 2xy + 3y^2$. The xy-plane is horizontal, and gravity pulls in the $-\hat{z}$ direction. Hence, the gravitational potential energy is $U(x, y) = x^2 - 2xy + 3y^2$ (with x and y measured in metres and U measured in units of mg). Suppose that other mysterious forces are at work on the ball, such that its trajectory is confined to the closed curve $V(x, y) = x^2 - (2/3)xy + y^2 = 1$. We would like to find the two points of locally minimum gravitational potential energy (consistent with the constraint) at which the ball could come to rest.

   It can be helpful to visualize the problem:

```
U[x_, y_] := x^2 - 2 x*y + 3 y^2;
P1 = ContourPlot[U[x, y], {x, -2, 2}, {y, -2, 2}, Contours -> {0.1, 0.2, 0.5, 1.0, 2.0,
    5.0, 10.0, 20.0}, PlotLegends -> Automatic];
V[x_, y_] := x^2 - (2/3) x y + y^2
P2 = ContourPlot[V[x, y] == 1, {x, -2, 2}, {y, -2, 2}, ContourStyle -> {Directive[White,
    Thick, Dashed]}];
Show[P1, P2]
```



   (a) Let's group the horizontal coordinates into a row vector $r^T = (x\ y)$ and reexpress the gravitational potential energy (a quadratic form) as $U(r) = r^T A r$. What are the elements of the matrix A? The matrix is not unique, but you will find it convenient to choose a matrix A that is symmetric.

   (b) Now define a function
   $$L(r, \lambda) = r^T A r - \lambda(r^T B r - 1)$$
   in which the constraint has been added as a Lagrange multiplier term. What are the elements of B? I encourage you to chose B symmetric also.

   (c) Show explicitly that enforcing stationarity, viz. $\partial L/\partial r_k = \partial L/\partial \lambda = 0$, leads to $(A + A^T)r = \lambda(B + B^T)r$ and $r^T B r = 1$. Provided that A and B are symmetric, this is just the generalized eigenvalue problem $Ar = \lambda Br$ with a normalization condition.

   (d) Find the two independent eigenvalue/eigenvector pairs, $(\lambda_1, r^{(1)})$ and $(\lambda_2, r^{(2)})$. Be sure that the $r^{(k)}$ vectors have been properly rescaled to satisfy the constraint.

   (e) Show that the gravitational energy values at the two locally stable points are 3 and 3/4.

You may want to check your answer against the results of this Mathematica code listing:

```
r = {x, y}
A = {{1, -1}, {-1, 3}}; MatrixForm[A]
Expand[r . A . r] == U[x, y]
B = {{1, -1/3}, {-1/3, 1}}; MatrixForm[B]
Expand[r.B.r] == V[x, y]
sol = Eigensystem[{A, B}]
Eigensystem[Inverse[B].A] == sol
eval = First[sol]
evec = Last[sol]
n1 = evec[[1]].B.evec[[1]]
ev1 = evec[[1]]/Sqrt[n1]
n2 = evec[[2]].B.evec[[2]]
ev2 = evec[[2]]/Sqrt[n2]
ev1.B.ev1
ev2.B.ev2
ev1.A.ev1
ev2.A.ev2
Show[P1, P2, ListPlot[{ev1, ev2}, PlotStyle -> {Red, PointSize[Large]}]]
```

3. A particular quantum Hamiltonian has the following matrix representation, which involves two real-valued elements, ε_1 and ε_2, and one complex-valued element, Δ, all having units of energy:

$$H = \begin{pmatrix} \varepsilon_1 & \Delta^* \\ \Delta & \varepsilon_2 \end{pmatrix}.$$

   (a) Using the notation $\Delta_1 = \operatorname{Re}\Delta$ and $\Delta_2 = \operatorname{Im}\Delta$, rewrite the Hamiltonian in terms of the 2×2 identity matrix, I, and the three Pauli matrices, σ_x, σ_y, and σ_z:

$$H = \frac{1}{2}(\varepsilon_1 + \varepsilon_2)I + \frac{1}{2}(\varepsilon_1 - \varepsilon_2)\sigma_z + \Delta_1\sigma_x + \Delta_2\sigma_y.$$

   Complete this problem by hand, but also verify that *Mathematica* can carry out the same decomposition algorithmically:

```
H = { {Subscript[\[Epsilon], 1], \[CapitalDelta]\[Conjugate]}, {\[CapitalDelta],
    Subscript[\[Epsilon], 2]} };
MatrixForm[H]
basis = {IdentityMatrix[2], Splice[PauliMatrix @ Range[3]]};
MatrixForm /@ basis
Tr[basis[[#]].H]/2 & /@ Range[4]
FullSimplify[(Tr /@ (basis.H/2)).basis] // MatrixForm
Assuming[
    Subscript[\[CapitalDelta], 1] \[Element] Reals &&
    Subscript[\[CapitalDelta], 2] \[Element] Reals, Simplify /@
    Refine[
        (Tr /@ (basis . H/2)) /. \[CapitalDelta] -> Subscript[\[CapitalDelta], 1]
            + I Subscript[\[CapitalDelta], 2]
    ].{\[DoubleStruckOne], Subscript[\[Sigma], x], Subscript[\[Sigma], y],
        Subscript[\[Sigma], z]}
]
```

(b) Show that quantum evolution operator associated with H has the form

$$e^{-itH/\hbar} = e^{-i\omega_1 t}\Big[I\cos\omega_2 t - \frac{it}{\hbar}\Big(\frac{1}{2}(\varepsilon_1 - \varepsilon_2)\sigma_z + \Delta_1\sigma_x + \Delta_2\sigma_y\Big)\operatorname{sinc}\omega_2 t\Big],$$

where the angular frequencies are defined as

$$\omega_1 = (\varepsilon_1 + \varepsilon_2)/2\hbar \quad \text{and} \quad \omega_2 = \frac{1}{\hbar}\sqrt{\Delta_1^2 + \Delta_2^2 + \frac{1}{4}(\varepsilon_1 - \varepsilon_2)^2}.$$

(c) Consider the special case in which $\varepsilon_1 = \varepsilon_2$ (the two energy levels coincide) and $\Delta = \Delta_1 > 0$ is purely real and positive. (i) Show that if the system is prepared in the state

$$\psi(0) = \begin{pmatrix} 1 \\ 0 \end{pmatrix}$$

at time zero then the wave function at all subsequent times is given by

$$\psi(t) = e^{-itH/\hbar}\psi(0) = e^{-i\varepsilon_1 t/\hbar}\begin{pmatrix} \cos(t\Delta_1/\hbar) \\ -i\sin(t\Delta_1/\hbar) \end{pmatrix}.$$

(ii) Verify that $\psi^\dagger\psi = 1\ \forall t$ and that the probabilities of measuring the system in levels 1 and 2 are

$$p_1(t) = \psi^\dagger\begin{pmatrix} 1 & 0 \\ 0 & 0 \end{pmatrix}\psi = \cos^2\Big(\frac{t\Delta_1}{\hbar}\Big) \quad \text{and} \quad p_2(t) = \psi^\dagger\begin{pmatrix} 0 & 0 \\ 0 & 1 \end{pmatrix}\psi = \sin^2\Big(\frac{t\Delta_1}{\hbar}\Big).$$